

Лабораторная работа № 2

ЭЛЕКТРОННЫЕ ТАБЛИЦЫ: ФОРМУЛЫ С ПРОВЕРКОЙ УСЛОВИЙ

Цель работы: изучить способы проверки условий с помощью формул, научиться производить вычисления с использованием формул с проверкой условий.

Задачи работы

1. Изучить основные способы проверки условий с помощью формул в *MS Excel*.
2. Научиться выполнять проверку простого условия и нескольких условий.
3. Изучить способ проверки совпадения результатов двух условий.
4. Научиться выполнять суммирование всех значений, которые соответствуют одному или нескольким условиям.
5. Научиться вычислять среднее для всех значений, отвечающих одному или нескольким условиям.
6. Научиться вычислять количество значений, отвечающих одному или нескольким условиям.

Перечень обеспечивающих средств

Задания лабораторной работы выполняются в электронной таблице *MS Excel* 2016 и выше.

Общие теоретические сведения

В *Excel* имеется несколько функций рабочего листа, предназначенных для проверки условий.

Проверка условий с помощью формул

Под *условием* понимается результат вычисления выражения, которое возвращает одно из двух значений: ИСТИНА или ЛОЖЬ. Исходя из вычисленного

значения условия, в формуле затем будет использоваться один из двух возможных вариантов выполнения вычислений. Иначе говоря, когда условие возвращает значение ИСТИНА, возвращается одно значение или вычисляется первое выражение, а остальная часть формулы просто игнорируется. Если же условие возвращает значение ЛОЖЬ, все происходит наоборот: возвращается другое значение или вычисляется второе выражение, а первая часть формулы игнорируется.

Проверка выполнения простого условия

На рис. 1 приведены список из нескольких стран Европы и соответствующие цены на газ за шесть месяцев. Для каждого значения цены в этой таблице требуется установить, выше она или ниже среднего значения цены на газ для всех указанных стран в данном месяце. Если цена выше средней, должно выводиться сообщение «Высокая», а если ниже средней – «Низкая». Для вывода этих сообщений будут использоваться ячейки, расположенные ниже исходной таблицы. Вот формула, которую можно использовать для решения поставленной задачи:

=ЕСЛИ(С3>СРЗНАЧ(С\$3:С\$11);"Высокая";"Низкая")

Функция ЕСЛИ – это фундаментальная функция проверки условий в *Excel*. Она имеет три аргумента: условие, действие, когда условие выполняется, и действие, когда условие не выполняется.

В примере аргумент условия имеет вид С3>СРЗНАЧ(С\$3:С\$11).

Выражения в аргументах задания условия всегда должны быть построены так, чтобы возвращать одно из двух возможных значений: ИСТИНА или ЛОЖЬ. Это означает, что в нем должен присутствовать оператор сравнения (например, знак равенства или знак «больше чем») либо другая функция рабочего листа, возвращающая значение ИСТИНА или ЛОЖЬ (например, ЕОШ или ЕПУСТО). В примере условие включает знак «больше чем» и предполагает сравнение значения в ячейке С3 со средним значением для содержимого всех ячеек диапазона С3:С11.

Если при проверке условия возвращается значение ИСТИНА (когда значение в ячейке C3 действительно больше среднего значения для содержимого ячеек заданного диапазона), в ячейку C14 с формулой возвращается второй аргумент функции ЕСЛИ – строка «Высокая».

В ячейке C15 с тем же средним значением сравнивается значение в ячейке C4. Поскольку оно меньше среднего, при проверке условия возвращается значение ЛОЖЬ, а значит, функция ЕСЛИ возвращает свой третий аргумент – строку «Низкая».

=ЕСЛИ(C3>СРЗНАЧ(C\$3:C\$11);"Высокая";"Низкая")						
В	С	Д	Е	Ф	Г	Н
Штат	Август 2000	Сентябрь 2000	Октябрь 2000	Ноябрь 2000	Декабрь 2000	Январь 2000
Германия	3,919	3,989	3,829	3,641	3,642	3,666
Франция	3,569	3,582	3,410	3,231	3,122	3,238
Испания	3,614	3,558	3,388	3,377	3,516	3,486
Португалия	3,761	3,703	3,518	3,419	3,520	3,527
Италия	3,577	3,540	3,318	3,143	3,113	3,272
Венгрия	3,933	3,879	3,700	3,633	3,736	3,734
Австрия	3,542	3,512	3,317	3,231	3,281	3,336
Великобритания	3,509	3,383	3,180	3,104	3,171	3,187
Швейцария	3,855	3,767	3,567	3,373	3,348	3,366
Германия	Высокая	Высокая	Высокая	Высокая	Высокая	Высокая
Франция	Низкая	Низкая	Низкая	Низкая	Низкая	Низкая
Испания	Низкая	Низкая	Низкая	Высокая	Высокая	Высокая
Португалия	Высокая	Высокая	Высокая	Высокая	Высокая	Высокая
Италия	Низкая	Низкая	Низкая	Низкая	Низкая	Низкая
Венгрия	Высокая	Высокая	Высокая	Высокая	Высокая	Высокая
Австрия	Низкая	Низкая	Низкая	Низкая	Низкая	Низкая
Великобритания	Низкая	Низкая	Низкая	Низкая	Низкая	Низкая
Швейцария	Высокая	Высокая	Высокая	Высокая	Низкая	Низкая

Рис. 1. Цены на газ в отдельных государствах по месяцам и их оценка

Проверка нескольких условий

Простые условия, подобные приведенному на рис. 1, можно связывать между собой; такой подход известен как «вложение функций». В этом случае аргументы «если условие выполнено» и «если условие не выполнено» также могут включать в себя проверку простых условий. В результате появляется возможность проверить более одного условия, причем выбор последующих проверяемых условий зависит от результата проверки первого условия.

На рис. 2 показан рабочий лист с двумя полями ввода, значение в которых выбирается пользователем: с полем *Тип автомобиля* и полем *Особенности*, предназначенным для уточнения характеристик автомобиля выбранного типа. Возможные значения для поля ввода *Особенности* приведены ниже в двух диапазонах – для каждого из двух допустимых типов автомобиля. Если пользователь выберет определенные значения в поле типа и в поле особенностей автомобиля, в поле *Какое авто* с помощью приведенной ниже формулы будет отображено название класса автомобилей с указанными характеристиками: Купе, Седан, Пикап или Трейлер.

=ЕСЛИ(E2="Легковой";ЕСЛИ(E3="2-х дверный";"Купе";"Седан");ЕСЛИ(E3="Есть кузов";"Пикап";"Трейлер"))

fx =ЕСЛИ(E2="Легковой";ЕСЛИ(E3="2-х дверный";"Купе";"Седан");ЕСЛИ(E3="Есть кузов";"Пикап";"Трейлер"))									
	D	E	F	G	H	I	J	K	L
Тип автомобиля:		Грузовик		Какое авто:	Пикап				
Особенности:		Есть кузов							
		Легковой							
		2-х дверный							
		4-х дверный							
		Грузовик							
		Есть кузов							
		Нет кузова							

Рис. 2. Определение класса автомобилей по заданным характеристикам

При выполнении проверки условий результат проверки первого условия определяет второе проверяемое условие. В нашем случае, если первое условие возвращает результат ИСТИНА, т. е. в ячейке E2 выбрано значение «Легковой», то вторым условием будет проверка, какое значение находится в ячейке E3: 2-х дверный или 4-х дверный. Если же в ячейке E2 выбрано значение «Грузовик», то вторым условием станет проверка, находится ли в ячейке E3 значение «Есть кузов» или значение «Нет кузова».

Как было показано выше, для выполнения проверки условий Excel предлагает функцию ЕСЛИ. Также, когда необходимо проверить более одного условия, можно использовать вложенные функции ЕСЛИ, т. е. в первой функции ЕСЛИ

как аргумент можно использовать другие функции ЕСЛИ. В примере первая функция ЕСЛИ проверяет значение в ячейке *E2*, но вместо того, чтобы просто вернуть значение, определяемое вторым аргументом, если результат первой проверки есть ИСТИНА, в формуле используется другая функция ЕСЛИ, которая проверяет значение в ячейке *E3*. Аналогичным образом третий аргумент является не просто конкретным значением, возвращаемым, когда результат первой проверки будет ЛОЖЬ, а третьей функцией ЕСЛИ, которая также проверяет значение в ячейке *E3*.

На рис. 2 пользователь выбрал в ячейке *E2* значение *Грузовик*. В результате в формуле в ячейке *H2* первая функция ЕСЛИ возвращает значение ЛОЖЬ, поскольку значение в ячейке *E2* не соответствует строке «Легковой», и на следующем этапе выполняется проверка, задаваемая третьим ее аргументом. В этом аргументе задана функция ЕСЛИ, сравнивающая значение в ячейке *E3* со строкой «Есть кузов». В примере результат сравнения – ИСТИНА, поэтому вся формула возвращает значение *Пикап*. Однако, если пользователь в ячейке *E3* выберет вариант *Нет кузова*, результатом второй проверки будет ЛОЖЬ и вся формула возвратит значение третьего аргумента второй функции ЕСЛИ, т. е. *Трейлер*.

Проверка совпадения результатов двух условий

Функции проверки условия, помимо вложения друг в друга, можно одновременно анализировать как аргументы в рамках одной функции И. Такой подход полезен, если одновременно необходимо проанализировать выполнение двух и более условий, прежде чем можно будет решить, по какой ветви далее пойдет выполнение расчетов.

На рис. 3 представлен перечень хранящихся на складе изделий с указанием их количества в наличии и размера скидки, установленной при их продаже. Обозначение каждого изделия состоит из трех элементов, разделенных дефисами.

Первый элемент определяет цех-изготовитель, второй указывает, чем это изделие является: отдельной деталью (*PRT*), сборочным блоком (*SUB*) или полностью собранным продуктом (*FIN*). Третий элемент представляет собой уникальный четырехзначный код. Задача состоит в том, чтобы назначить скидку в

10 % только для тех изделий, которые изготовлены в цеху 202 и являются полностью собранным продуктом. Для всех остальных изделий на складе скидка отсутствует. Вот как выглядит формула, позволяющая решить поставленную задачу.

=ЕСЛИ(И(ЛЕВСИМВ(B3;3)="202";ПСТР(B3;5;3)="FIN");10%;0%)

Функция ЕСЛИ возвращает значение 10 %, если проверка условия дает значение ИСТИНА, и 0 %, если результат проверки условия – ЛОЖЬ. В ее аргументе, представляющем проверяемое условие (первый аргумент), требуется записать выражение, которое вернет значение ИСТИНА только в том случае, когда первый элемент кода изделия равен 202, а второй имеет значение *FIN*. Для решения подобных задач *Excel* предлагает функцию И. Эта функция допускает передачу ей на вход до 255 логических аргументов, разделенных символами точки с запятой. Логические аргументы представляют собой выражения, возвращающие одно из двух значений: ИСТИНА или ЛОЖЬ. В примере используются всего два логических аргумента.

Первый логический аргумент, ЛЕВСИМВ(B5;3) ="202", возвращает значение ИСТИНА, если первые три символа кода изделия в ячейке B3 равны 202. Второй логический аргумент, ПСТР(B5;5;3) ="FIN", возвращает значение ИСТИНА в том случае, если три символа в коде изделия, начиная с пятой позиции, будут иметь значение *FIN*.

Функция и вернет значение истина только в том случае, если все ее логические аргументы вернут значение ИСТИНА. Если хотя бы один из логических аргументов функции и вернет значение ЛОЖЬ, функция также вернет значение ЛОЖЬ.

В ячейке D3 первый логический аргумент функции и вернет значение ИСТИНА, поскольку первые три символа кода изделия равны 202. Второй логический аргумент вернет значение ЛОЖЬ, поскольку средний элемент кода изделия – *PRT*, а не *FIN*.

Таким образом, в ячейке D5 формула вернет значение 10 %, поскольку оба логических аргумента функции и вернут значение истина.

=ЕСЛИ(И(ЛЕВСИМВ(B3;3)="202";ПСТР(B3;5;3)="FIN");10%;0%)							
B	C	D	E	F	G	H	I
Код изделия	Количество	Скидка					
202-PRT-3013	76	0%					
201-FIN-1452	69	0%					
202-FIN-8206	12	10%					
201-FIN-8238	79	0%					
203-FIN-8882	16	0%					
202-PRT-9587	87	0%					
203-FIN-4614	97	0%					
201-PRT-2478	25	0%					
202-SUB-1955	14	0%					
201-SUB-8641	67	0%					
202-FIN-9069	40	10%					
202-PRT-7937	61	0%					
201-SUB-3124	70	0%					
203-SUB-4369	16	0%					
202-FIN-6273	74	10%					
203-SUB-3972	85	0%					
203-PRT-3335	84	0%					
201-SUB-1022	48	0%					
203-FIN-3507	17	0%					
203-SUB-8304	31	0%					

Рис. 3. Список изделий на складе

Проверка удовлетворения первого или второго условия

В предыдущем примере скидка предоставлялась на определенный продукт, определяемый по его коду изделия. В следующем примере требуется расширить количество продуктов, на которые предоставляется скидка. Как и раньше, скидка предоставляется только на готовые продукты, но количество цехов увеличивается до двух: цех 202 и цех 203. На рис. 4 представлен перечень продуктов на складе с процентом скидки, установленным по новым правилам, – для этой цели использовалась следующая формула:

=ЕСЛИ(И(ИЛИ(ЛЕВСИМВ(B3;3)="202";ЛЕВСИМВ(B3;3)="203");
ПСТР(B3;5;3)="FIN");10%;0%)

=ЕСЛИ(И(ИЛИ(ЛЕВСИМВ(В3;3)="202";ЛЕВСИМВ(В3;3)="203");ПСТР(В3;5;3)="FIN");10%;0%)										
В	С	Д	Е	Ф	Г	Н	І	Ј	К	І
Код изделия	Количество	Скидка								
202-PRT-3013	76	0%								
201-FIN-1452	69	0%								
202-FIN-8206	12	10%								
201-FIN-8238	79	0%								
203-FIN-8882	16	10%								
202-PRT-9587	87	0%								
203-FIN-4614	97	10%								
201-PRT-2478	25	0%								
202-SUB-1955	14	0%								
201-SUB-8641	67	0%								
202-FIN-9069	40	10%								
202-PRT-7937	61	0%								
201-SUB-3124	70	0%								
203-SUB-4369	16	0%								
202-FIN-6273	74	10%								
203-SUB-3972	85	0%								
203-PRT-3335	84	0%								
201-SUB-1022	48	0%								
203-FIN-3507	17	10%								
203-SUB-8304	31	0%								

Рис. 4. Скидка на изделия предоставляется по новым правилам

В этой формуле аргумент условия функции ЕСЛИ был расширен для учета изменений в правилах предоставления скидки. Функция И является ограничивающей, поскольку все ее аргументы должны иметь значение ИСТИНА, чтобы эта функция также вернула значение ИСТИНА. В противоположность ей функция ИЛИ является включающей, так как она возвращает значение ИСТИНА, если хотя бы один из ее аргументов имеет значение ИСТИНА.

В примере функция ИЛИ была вложена в функцию И, – она используется в качестве одного из аргументов внешней функции ИЛИ.

На рис. 4 значение в ячейке D9 демонстрирует, что изделие, на которое раньше не было скидки, теперь ее получило в связи с введением новых правил ее предоставления. В данном случае выражение с функцией ИЛИ, а именно – ИЛИ(ЛЕВСИМВ(В3;3)="202"/ЛЕВСИМВ(В3;3)="203"), возвращает значение ИСТИНА, поскольку один из его аргументов также имеет значение ИСТИНА.


Вычисления с использованием формул с проверкой условий

Простые функции проверки условий, подобные функции ЕСЛИ, обычно работают только с одним значением или ячейкой за раз. Поэтому *Excel* предоставляет также несколько других функций проверки условий, предназначенных специально для обобщения данных, например, суммирования или вычисления среднего значения.

Суммирование всех значений, которые отвечают определенному условию

На рис. 5 представлен список бухгалтерских счетов с положительным и отрицательным балансами. Необходимо просуммировать все отрицательные остатки, а затем сравнить полученное значение с суммой всех положительных остатков, чтобы убедиться, что они равны. Для суммирования значений с учетом заданного условия *Excel* предоставляет функцию СУММЕСЛИ:

=СУММЕСЛИ(C3:C12;"<0")



В	С
Счет	Баланс
1510 Оборудование	9 863,00
1540 Накопленная амортизация	-9 502,00
1690 Земля	5 613,00
1915 Прочие активы	8 653,00
2320 Заработная плата	-6 937,00
2420 Текущая часть долгосрочной задолженности	-6 826,00
2440 Депозиты от клиентов	-3 717,00
5800 Стоимость проданных товаров, другое	73,00
5900 Возврат покупок и пособия	-4 443,00
6300 Расходы на благотворительные взносы	7 223,00
Отрицательные балансы	-31 425,00
Положительные балансы	31 425,00

Рис. 5. Суммирование отрицательных значений

Функция СУММЕСЛИ извлекает значение из каждой ячейки в диапазоне C3:C12 и проверяет его на соответствие заданному условию (второй аргумент функции). Если анализируемое значение меньше нуля, оно соответствует условию и включается в сумму, возвращаемую в качестве результата. Если же значение равно или больше нуля, оно игнорируется. Текстовые значения и пустые

ячейки также игнорируются. В примере, представленном на рис. 5, первым проверяется значение в ячейке C3. Поскольку оно больше нуля, функция его игнорирует. Следующим проверяется значение в ячейке C4. Оно отвечает заданному условию, поскольку является отрицательным, а значит, добавляется к общей сумме. Тот же алгоритм применяется ко всем оставшимся ячейкам диапазона. В конечном счете в общую сумму будут включены значения в ячейках C4, C7, C8, C9 и C11, а значения в остальных ячейках диапазона будут проигнорированы.

Функция СУММЕСЛИ имеет третий необязательный аргумент, называемый диапазон_суммирования. До этого момента в формуле проверка условия выполнялась непосредственно для тех числовых значений, которые суммировались. Если воспользоваться третьим аргументом функции СУММЕСЛИ, то можно дополнительно указать диапазон ячеек, в которых будут находиться суммируемые значения, тогда как проверка условий будет выполняться для значений ячеек другого диапазона, заданного первым аргументом. На рис. 6 показан список объемов продаж в различных регионах. В этом случае, чтобы определить общую сумму продаж по региону *Восток*, можно воспользоваться следующей формулой:

=СУММЕСЛИ(B2:B11;"Восток";C2:C11)

=СУММЕСЛИ(B2:B11;"Восток";C2:C11)	
В	С
Центр	2 714 745,31
Центр	1 434 322,83
Север	1 983 811,70
Восток	929 430,78
Восток	3 154 066,47
Центр	1 264 430,64
Север	4 674 274,42
Восток	940 684,25
Центр	2 497 381,24
Центр	1 728 260,75
Сумма по региону Восток	5 024 181,50
Сумма по регионам Север и Центр	16 297 226,89


Рис. 6. Объем продаж по региону Восток

В качестве второго аргумента можно использовать функцию СЕГОДНЯ (для получения текущей даты) или большинство других функций Excel. На рис. 7 отображен список дат и числовых значений. Чтобы просуммировать числовые значения, соответствующие текущей дате, нужно использовать формулу:

=СУММЕСЛИ(B3:B11;СЕГОДНЯ();C3:C11)

Чтобы просуммировать числовые значения, соответствующие сегодняшней и более ранним датам, соедините символом конкатенации выражение "<=" и функцию СЕГОДНЯ:

=СУММЕСЛИ(B3:B11;"<=СЕГОДНЯ()";C3:C11)



The screenshot shows an Excel spreadsheet with the following data:

В	С	Д	Е	Ф
Дата	Число			
19.01.2019	10			
20.01.2019	20			
21.01.2019	30			
22.01.2019	40			
23.01.2019	50			
24.01.2019	60			
25.01.2019	70			
26.01.2019	80			
27.01.2019	90			
Сумма за сегодня	0			
Сумма за сегодня и ранее	450			

The formula bar at the top shows: =СУММЕСЛИ(B3:B11;СЕГОДНЯ();C3:C11)

Рис. 7. Использование функции СЕГОДНЯ в условии для функции СУММЕСЛИ

В записи аргумента условия для функции СУММЕСЛИ можно использовать два подстановочных символа для поиска по шаблону. Знак вопроса (?) представляет любой одиночный символ, а звездочка (*) – произвольное количество любых символов (ни одного, один, два и т. д.):

=СУММЕСЛИ(B2:B11;"?e*";C2:C11)

Формула выше предполагает суммирование значений, хранящихся в ячейках C2:C11, если соответствующие им ячейки в диапазоне B2:B11 содержат текстовое значение длиной два и более знаков, вторым символом в котором является

строчная буква «е». Применив эту формулу к данным, представленным на рис. 6, в ячейке C15 мы получим сумму продаж по регионам Север и Центр, поскольку в обоих этих названиях второй буквой является «е», а в слове Восток – нет.

Суммирование значений, которые соответствуют двум и более условиям

Ограничением на использование функции СУММЕСЛИ, представленной на рис. 6, является тот факт, что она работает только с одним условием. Если необходимо проанализировать сразу несколько условий, следует обратиться к функции СУММЕСЛИМН.

На рис. 8 показана часть списка стран с указанием их валового внутреннего продукта (ВВП) за период с 2000 по 2009 год. Предположим, необходимо узнать суммарный ВВП за период с 2003 по 2006 год. Для подсчета этого значения используется функция рабочего листа *Excel* СУММЕСЛИМН, позволяющая проверить два и более условий, в примере – значения ячеек в столбцах Страна и Год.

=СУММЕСЛИМН(D3:D212;B3:B212;G3;C3:C212;">="&G4;C3:C212;"<="&G5)					
В	С	Д	Е	Ф	Г
Страна	Год	ВВП			
Австралия	2000	399 594		Страна	Австралия
Австралия	2001	377 207		Начальный год	2003
Австралия	2002	423 676		Конечный год	2006
Австралия	2003	539 162			
Австралия	2004	654 968		Суммарный ВВП	2 702 792
Австралия	2005	730 729			
Австралия	2006	777 933			
Австралия	2007	945 364			
Австралия	2008	1 051 261			
Австралия	2009	993 349			
Бельгия	2000	233 354			
Бельгия	2001	232 686			
Бельгия	2002	253 689			
Бельгия	2003	312 285			

Рис. 8. Определение суммы значений по трем условиям

В примере используется функция:

=СУММЕСЛИМН(D3:D212;B3:B212;G3;C3:C212;">="&G4;C3:C212;"<="&G5)

Список аргументов функции СУММЕСЛИМН начинается с диапазона ячеек, содержащих значения, которые следует суммировать (D3:D212). Остав-

шиеся аргументы следуют парами в виде проверяемый_диапазон, критерий. Благодаря такой организации списка аргументов у функции СУММЕСЛИМН всегда должно быть нечетное количество аргументов. Пара аргументов для первого критерия является обязательной. Все остальные пары аргументов являются необязательными.

В примере содержимое каждой ячейки в диапазоне D3:D212 добавляется к общему итогу только в том случае, если значения в ячейках диапазонов B3:B212 и C3:C212 соответствуют указанным условиям. Условием для диапазона B3:B212 является соответствие значению, содержащемуся в ячейке G3. В примере используются два условия отбора по году, поскольку заданы верхняя и нижняя границы анализируемого периода. Нижняя граница определяется в ячейке G4, а верхняя – в ячейке G5.

В условиях отбора по году адреса этих двух ячеек соединяются оператором конкатенации с выражениями «больше или равно» и «меньше или равно» соответственно.

Содержимое ячейки будет добавлено к общей сумме только в том случае, если все три условия выполняются.

Подсчет значений, которые отвечают заданному условию

Суммирование значений – это не единственный способ обобщения данных, доступный в Excel. Подобно функциям суммирования СУММЕСЛИ и СУММЕСЛИМН, Excel предоставляет функции подсчета количества значений в диапазоне ячеек, отвечающих заданным условиям.

На рис. 9 показана часть списка стран с указанием их валового внутреннего продукта (ВВП) за период с 2000 по 2009 год. Предположим, нас интересует, как часто ВВП разных стран был больше или равен одному миллиону. Если этот критерий ввести в ячейку G3, то требуемая формула будет иметь следующий вид:

=СЧЁТЕСЛИ(D3:D212;G3)

=СЧЁТЕСЛИ(D3:D212;G3)					
В	С	Д	Е	Ф	Г
Страна	Год	ВВП			
Австралия	2000	399 594		Критерий для количества	>=1000000
Австралия	2001	377 207			
Австралия	2002	423 676		Количество	96
Австралия	2003	539 162			
Австралия	2004	654 968		ВВП между	300000
Австралия	2005	730 729		И	1000000
Австралия	2006	777 933		Год	2002
Австралия	2007	945 364			
Австралия	2008	1 051 261		Количество по трем условиям	9
Австралия	2009	993 349			
Бельгия	2000	233 354			
Бельгия	2001	232 686			
Бельгия	2002	253 689			

Рис. 9. Определение количества значений по условию

Принцип работы функции СЧЁТЕСЛИ подобен принципу работы функции СУММЕСЛИ, представленной на рис. 5. Единственным заметным различием является то, что данная функция подсчитывает количество значений, соответствующих заданному критерию, вместо суммирования этих значений. Еще одним отличием является отсутствие третьего необязательного аргумента, который есть у функции СУММЕСЛИ.

В формуле примера используется несколько иной способ задания аргумента критерия в ячейке G3. Конкатенация строковых значений выполнена непосредственно в ячейке G3, а не во втором аргументе функции: ">="&10^6. Для представления значения в один миллион используется оператор возведения в степень (^). Представление больших значений с использованием оператора возведения в степень помогает уменьшить количество ошибок, связанных с введением неправильного количества нулей.

Подсчет количества значений, отвечающих двум и более условиям

Функция СУММЕСЛИМН имеет свой аналог для вычисления количества значений, отвечающим двум и более условиям – функцию СЧЁТЕСЛИМН.

На рис. 10 показан пример использования функции СЧЁТЕСЛИМН для определения количества стран, у которых значение ВВП находилось в диапазоне от 300 000 до 1 000 000 в 2002 году. Значения ВВП, используемые в качестве аргумента, указаны в ячейках G7 и G8, значение года – в ячейке G9.

=СЧЁТЕСЛИМН(D3:D212;">="&G7;D3:D212;"<="&G8;C3:C212;G9)						
В	С	Д	Е	Ф	Г	
Страна	Год	ВВП				
Австралия	2000	399 594		Критерий для количества	>=1000000	
Австралия	2001	377 207				
Австралия	2002	423 676		Количество	96	
Австралия	2003	539 162				
Австралия	2004	654 968		ВВП между	300000	
Австралия	2005	730 729		И	1000000	
Австралия	2006	777 933		Год	2002	
Австралия	2007	945 364				
Австралия	2008	1 051 261		Количество по трем условиям	9	
Австралия	2009	993 349				
Бельгия	2000	233 354				
Бельгия	2001	232 686				
Бельгия	2002	253 689				
Бельгия	2003	312 285				

Рис. 10. Определение количества значений по трем условиям

В примере используется формула:

=СЧЁТЕСЛИМН(D3:D212;">="&G7;D3:D212;"<="&G8;C3:C212;G9)

Аргументы функции СЧЁТЕСЛИМН следуют парами в виде диапазон_условия, условие. Благодаря такой организации списка аргументов у функции СЧЁТЕСЛИМН всегда должно быть четное количество аргументов. Пара аргументов для первого критерия является обязательной. Все остальные пары аргументов являются необязательными.

В первой паре значения из диапазона D3:D212 проверяются на истинность по условию «больше или равно» значению в ячейке G7. Аналогично во второй паре значения из того же диапазона проверяются на истинность по условию «меньше или равно» значению в ячейке G8. В третьей паре проверяется истинность, при совпадении значений в диапазоне C3:C212 со значением в ячейке G9.

Таким образом, по формуле будет найдено количество стран, в которых в 2002 году ВВП находилось в диапазоне от 300 тысяч до 1 миллиона.

Вычисление среднего для всех значений, отвечающих определенному условию

После суммирования и подсчета строк вычисление среднего значения для заданного набора числовых значений является следующей по частоте использования операцией обобщения данных. Среднее значение, которое иначе называют арифметическим средним, представляет собой сумму этих значений, деленную на их количество.

На рис. 11 показана часть списка стран с указанием их валового внутреннего продукта (ВВП) за период с 2000 по 2009 год. Предположим, нас интересует, среднее значение ВВП за весь период по одной из стран. Если этот критерий (название страны) ввести в ячейку G3, то требуемая формула будет иметь следующий вид:

=СРЗНАЧЕСЛИ(B3:B212;G3;D3:D212)

=СРЗНАЧЕСЛИ(B3:B212;G3;D3:D212)					
В	С	Д	Е	Ф	Г
Страна	Год	ВВП			
Австралия	2000	399 594		Критерий	Австралия
Австралия	2001	377 207			
Австралия	2002	423 676		Среднее значение	689324,3
Австралия	2003	539 162		за период 2000-2009 гг.	
Австралия	2004	654 968			
Австралия	2005	730 729			
Австралия	2006	777 933			
Австралия	2007	945 364			
Австралия	2008	1 051 261			
Австралия	2009	993 349			
Бельгия	2000	233 354			
Бельгия	2001	232 686			

Рис. 11. Определение среднего арифметического значения по условию

Принцип работы функции СРЗНАЧЕСЛИ подобен принципу работы функции СУММЕСЛИ, представленной на рис. 5. Единственным различием является

то, что данная функция подсчитывает среднее арифметическое значений, соответствующих заданному критерию, вместо суммирования этих значений.

Вычисление среднего для всех значений, отвечающих двум и более условиям

Вместе с функциями СУММЕСЛИМН и СЧЁТЕСЛИМН пользователям Excel доступна функция СРЗНАЧЕСЛИМН, позволяющая вычислить среднее арифметическое для всех значений из диапазона, отвечающих двум и более условиям.

На рис. 12 показана часть списка стран с указанием их валового внутреннего продукта (ВВП) за период с 2000 по 2009 год. Предположим, необходимо узнать среднее арифметическое значение ВВП Австралии за период с 2003 по 2006 год. Для подсчета этого значения используется функция рабочего листа Excel СРЗНАЧЕСЛИМН, позволяющая проверить два и более условий, в примере – значения ячеек в столбцах Страна и Год.

=СРЗНАЧЕСЛИМН(D3:D212;B3:B212;G3;C3:C212;">="&G8;C3:C212;"<="&G9)							
В	С	Д	Е	Ф	Г	Н	И
Страна	Год	ВВП					
Австралия	2000	399 594		Критерий	Австралия		
Австралия	2001	377 207					
Австралия	2002	423 676		Среднее значение	689324,3		
Австралия	2003	539 162		за период 2000-2009 гг.			
Австралия	2004	654 968					
Австралия	2005	730 729		Период с	2003		
Австралия	2006	777 933		по	2006		
Австралия	2007	945 364					
Австралия	2008	1 051 261		Среднее значение			
Австралия	2009	993 349		за период	675698		
Бельгия	2000	233 354					
Бельгия	2001	232 686					

Рис. 12. Определение среднего арифметического значения по трем условиям

В примере используется функция:

=СРЗНАЧЕСЛИМН(D3:D212;B3:B212;G3;C3:C212;">="&G8;C3:C212;
"<="&G9)

Функция СРЗНАЧЕСЛИМН по своей структуре подобна рассмотренной выше функции СУММЕСЛИМН. Первый аргумент определяет диапазон значений для вычисления среднего, а остальные до 127 пар аргументов определяют анализируемый диапазон и критерий отбора.

Задания

Во всех заданиях в ячейках *Excel* должны быть указаны не только исходные данные и результаты, но и пояснительные данные (например, названия столбцов, строк, наименование результата и т. п.).

Задание 1. Перейти за сайт Новосибирскстата в раздел «Всероссийская сельскохозяйственная перепись-2016» по адресу: <https://54.rosstat.gov.ru/folder/36883>. Скачать файл «Новосибирская область: Том 1. Основные итоги Всероссийской сельскохозяйственной переписи 2016 года. Книга 2. Основные итоги Всероссийской сельскохозяйственной переписи 2016 года по муниципальным районам и городским округам Новосибирской области» в формате *Excel*.

В новой рабочей книге *Excel* переименовать Лист 1, указав в названии листа свою фамилию и номер варианта.

На листе разместить статистические исходные данные (2 показателя) на 1 июля 2016 года по любым 10 районам Новосибирской области согласно своему варианту.

В отдельном столбце вычислить отношение одной величины к другой. Результат представить в процентном формате с точностью до сотых долей.

Варианты выполнения задания.

1. Отношение площади ягодников к площади многолетних плодовых насаждений и ягодных культур.
2. Отношение площади пашни к площади сельскохозяйственных угодий.
3. Отношение площади сенокосов к площади сельскохозяйственных угодий.
4. Отношение площади пастбищ к площади сельскохозяйственных угодий.
5. Отношение площади многолетних насаждений к площади сельскохозяйственных угодий.
6. Отношение площади залежи к площади сельскохозяйственных угодий.
7. Отношение посевной площади пшеницы к общей посевной площади сельскохозяйственных культур под урожай.

8. Отношение посевной площади ржи к общей посевной площади сельскохозяйственных культур под урожай.

9. Отношение посевной площади ячменя к общей посевной площади сельскохозяйственных культур под урожай.

10. Отношение посевной площади овса к общей посевной площади сельскохозяйственных культур под урожай.

В столбце, расположенном правее от статистических данных, определить является ли вычисленное отношение по каждому району больше или меньше среднего медианного значения по всем 10 районам.

В следующем столбце определить является ли вычисленное отношение по каждому району больше или меньше среднего арифметического значения по всем 10 районам.

Сравнить полученные результаты. Совпадают ли они по районам или нет? Почему?

Задание 2. На листе 2 в столбце *B* составьте вымышленный список *ID* студентов разных групп, курсов и специальностей. *ID* должны составляться по шаблону: «номер (название) группы, дефис, пятизначный номер зачетки студента». Например, БК-21-45398 или ПГ-31-23095. Список должен включать не менее 24 *ID*. В списке обязательно должны быть несколько *ID* студентов групп БК и ПГ, 2 и 3 курсов.

Нужно указать напротив каждого *ID* студента, должен ли он проходить тестирование или нет. Тестирование проходят только студенты по следующему условию отбора: код специальности совпадает с БК или ПГ и курс 2 или 3.

Задание 3. На лист 3 в столбец *A* скопировать исходные данные (*ID* студентов) с листа 2. Выделить из *ID* номер (название) группы каждого студента и поместить эти данные в столбец *B*. В столбец *C* добавить вымышленные данные о размере стипендии. Через одну строку от последнего значения размера стипендии вычислить суммарный размер стипендии студентов группы БК-21 с помощью функции СУММЕСЛИ и данных из столбца *B*.

В следующей строке с помощью функции СУММЕСЛИМН вычислить суммарный размер стипендии по следующим критериям: студенты обучаются на 3 курсе и размер стипендии больше 2400. Номер курса определить с помощью подстановочных символов на основе номера (названия) группы (сформировать шаблон).

Задание 4. Задание выполняется на листе 4 по вариантам на основе статистических данных из официального издания «Россия и страны мира. 2020», размещенных на странице https://rosstat.gov.ru/bgd/regl/b20_39/Main.htm.

Варианты выполнения задания.

1. Производство плодов, ягод, цитрусовых и винограда.
2. Производство сахара-песка.
3. Производство растительных масел.
4. Производство макаронных изделий.
5. Производство хлопчатобумажных тканей.
6. Производство бумаги и картона.
7. Производство автомобильного бензина.
8. Производство минеральных удобрений.
9. Производство цемента.
10. Производство чугуна.

В данные, перенесенные в *Excel*, добавить дополнительный столбец «Часть света» для указания части света, в которой находится конкретная страна.

Россию отнести к части света Европа.

На основе исходных данных с помощью математических и статистических функций СУММ, СРЗНАЧ, СЧЁТ, НАИБОЛЬШИЙ, НАИМЕНЬШИЙ, МАКС, СУММЕСЛИ, СУММЕСЛИМН, СЧЁТЕСЛИ, СЧЁТЕСЛИМН, СРЗНАЧЕСЛИ, СРЗНАЧЕСЛИМН выполнить расчет статистических показателей:

1) суммарное значение за весь период (все годы) по каждой стране (использовать функцию СУММ);

2) среднее арифметическое значение за весь период (все годы) по каждой стране (использовать функцию СРЗНАЧ);

3) суммарное значение за каждый год по всем странам вместе (использовать функцию СУММ);

4) среднее арифметическое значение за каждый год по всем странам вместе (использовать функцию СРЗНАЧ);

5) суммарное значение за весь период (все годы) отдельно по странам Европы, Азии, Африки, Америки, Австралии и Океании (использовать функцию СУММЕСЛИ);

6) среднее арифметическое значение за весь период (все годы) отдельно по каждой части света: Европа, Азия, Африка, Америка, Австралия и Океания (использовать функцию СРЗНАЧЕСЛИ);

7) количество стран по каждой части света (использовать функцию СЧЁТЕСЛИ);

8) суммарное значение за 2013 год отдельно по странам Европы, Азии, Африки, Америки, Австралии и Океании, в которых значение показателя в этот год больше четвертого наибольшего значения по всем странам вместе (использовать функции СУММЕСЛИМН, НАИБОЛЬШИЙ);

9) среднее арифметическое значение за 2014 год отдельно по странам Европы, Азии, Африки, Америки, Австралии и Океании, в которых значение показателя в этот год меньше четвертого наименьшего значения по всем странам вместе (использовать функции СРЗНАЧЕСЛИМН, НАИМЕНЬШИЙ);

10) количество стран по каждой части света, в которых среднее арифметическое значение за весь период (все годы) больше среднего арифметического значения в соответствующей части света (использовать функцию СЧЁТЕСЛИМН).

По 6 показателю построить гистограмму, разместить на отдельном листе, указать подробное название диаграммы и добавить справа легенду. Сделать выводы по диаграмме.

Вычислить количество стран, название которых начинается на первую букву вашей фамилии, максимальное значение которых за весь период меньше

среднего арифметического значения по всем странам за весь период (использовать функции СЧЁТЕСЛИМН, МАКС; предварительно в отдельном столбце вычислить максимальное значение).

Если страны, названия которых начинаются на первую букву вашей фамилии отсутствуют, то использовать следующую по алфавиту букву.

Контрольные вопросы

1. К каким категориям относятся функции с проверкой условий?
2. Опишите работу функций ЕСЛИ, И, ИЛИ.
3. В чем особенность использования функций СУММЕСЛИ и СУММЕСЛИМН?
4. В чем особенность использования функций СРЗНАЧЕСЛИ и СРЗНАЧЕСЛИМН?
5. В чем особенность использования функций СЧЁТЕСЛИ и СЧЁТЕСЛИМН?

Содержание отчета

1. Титульный лист
2. Цель работы.
3. Номер варианта.
4. Формулировка задания.
5. Описание результатов выполнения задания, включая скриншоты области рабочих листов электронной таблицы с выполненными заданиями.
6. Подробный вывод о проделанной работе.